A TUTORIAL ON THE SAS® MACRO LANGUAGE
John J. Cohen

Abstract

The SAS Macro language is another language that rests on top of regular SAS code. If used properly, it can make programming easier and more fun. However, not every program is improved by using macros. Furthermore, it is another language syntax to learn, and can create problems in debugging programs that are even more entertaining than those offered by regular SAS.

We will discuss using macros as code generators, saving repetitive and tedious effort, for passing parameters through a program to avoid hard coding values, and to pass code fragments, thereby making certain tasks easier than using regular SAS alone. Macros facilitate conditional execution and can be used to create program modules that can be standardized and re-used throughout your organization. Finally, macros can help us create interactive systems in the absence of SAS/AF.

When we are done, you will know the difference between a macro, a macro variable, a macro statement, and a macro function. We will introduce interaction between macros and regular SAS language, offer tips on debugging macros, and discuss SAS macro options.

MACROS

Macros consist of collections of regular SAS program statements, macro variables, macro language statements, and macro functions contained within a %MACRO and a %MEND. The %MACRO statement includes a name and the macro is called using the macro’s name preceded by a %.

Figure 1 A Simple Macro

```sas
%MACRO MYFIRST;
PROC PRINT DATA=CLAIMS;
TITLE1 "TESTING MY FIRST MACRO";
%MEND MYFIRST;

%MYFIRST; /* the macro call */
```

Macro MYFIRST is not very interesting, but it is syntactically correct. The PROC PRINT statement will execute, once we call the macro, exactly as it would if there were no macro language statements.

SAS recognizes that this is a macro (and not a collection of regular SAS programming statements) because of the % symbol. Note that these macro statements are ended with semicolons (;), and the selection of double quotes (") for the TITLE statement is deliberate.

MACRO VARIABLES

Macro variables are indicated by preceding a variable name with an ampersand (&). Following regular SAS variable naming conventions, we will create a macro variable called &WHEN. We will assign a value to &WHEN when we call the macro. This value (“9409” in our example) will be substituted in the space taken up by the macro variable &WHEN at execution time. Previewing our discussion of macro options, with SYMBOLGEN on, your SAS log will note that:

SYMBOLGEN: Macro variable WHEN resolves to 9409

Figure 2 Passing Values With Macro Variables

```sas
%MACRO CLAIMREP(WHEN=);
DATA REPORT; SET CLAIMS;
IF DATE = &WHEN THEN
  CURRENT = AMOUNT;
ELSE IF DATE = (&WHEN - 1) THEN
  PRIOR = AMOUNT;
ELSE DELETE;
PROC PRINT DATA=REPORT;
  VAR CURRENT PRIOR;
TITLE1 "MONTHLY CLAIMS REPORT FOR &WHEN";
TITLE2 "AVOIDING ""HARD-CODING""";
%MEND CLAIMREP;

%CLAIMREP(WHEN=9809);
```

When macro CLAIMREP executes, observations from dataset CLAIMS will be checked for a value of DATE equal to the value we assigned to &WHEN (“9409”) or the prior month (&WHEN-1, which would resolve to “9408”). These will be output into dataset REPORT and printed. Note that we also use &WHEN in the TITLE statement, thereby
automatically documenting which month is contained in the report. We can submit next month’s program by simply changing the macro variable value in the call to:

%CLAIMREP (WHEN=9810);

We can verify last year’s report with:

%CLAIMREP (WHEN=9709);

This construct allows us to enter a value (using &WHEN) into the program just once, even though we use the value three times. In a complicated program with many referenced to a particular parameter value, date range, contract, company, or the like, the chances of forgetting or mistyping one are remarkably high.

MACRO STATEMENTS

Our first macro statements were the %MACRO and %MEND. Other macro statements, often similar to their regular SAS equivalents, also start with the % symbol, such as %DO, %END, %IF%THEN, %ELSE, %GLOBAL, %LOCAL, and %LET. In the following example, we will run an annualized version of %CLAIMREP, once for each of the last ten years. We will use the macro language to generate ten copies of the program.

Figure 3 Using Macro Statements To Generate Code

%MACRO CLAIMREP;
%LET CURYEAR = 98;
%DO YEAR = 89 %TO 98;
 %IF "&YEAR" = "&CURYEAR" %THEN %LET NOTE = FOOTNOTE "THRU &SYSDATE";
 PROC PRINT DATA = CLAIMS;
 WHERE YEAR = &YEAR;
 TITLE1 "ANNUAL REPORT FOR 19&YEAR";
 &NOTE;
 %END;
%MEND CLAIMREP;

%CLAIMREP;

In this example, the macro CLAIMREP will generate ten copies of the regular SAS program, one for each iteration of the %DO loop. Each copy will print a report for the year assigned to it by the %DO loop. The current year’s version (here set to “98”) will include a footnote describing the completeness of the data.

Note that we assign values to &CURYEAR and &NOTE using a %LET statement, and to &YEAR with a %DO statement (all without ampersands). &SYSDATE is an automatic macro variable created by SAS. In TITLE1, the phrase “19&YEAR” will resolve to “1989” the first time through the %DO loop, to “1990” for the second iteration, and on to “1998” in the last run through. Similarly, the WHERE statement will be comparing values of DATA step variable YEAR to a macro variable value of “89” the first time, “90” the second, and so on.

MACRO FUNCTIONS

Macro functions operate much as regular SAS functions, except that the arguments are within the context of the macro language. Most operate on character strings (e.g., %SUBSTR) or control the exact interpretation of macro special symbols (e.g., %STR). Macro variables are character variables so the %EVAL function may be required to perform arithmetic within macro statements. Figure 3 above included a simple instance of conditional execution (whether or not to create a value for the FOOTNOTE statement). In Figure 4 we will have the program elect to create a backup of the CLAIMS dataset if we are running the report at year’s end (i.e., month 12).

Figure 4 Macro Functions

%MACRO CLAIMREP(WHEN=); /* main macro */
%GLOBAL COUNT TOTAL;
%IF %SUBSTR(&WHEN,3,2) = 12
 %THEN %BACKITUP;

DATA REPORT;
 SET CLAIMS END=EOF;
 RETAIN COUNT TOTAL 0;
 IF DATA = &WHEN THEN DO;
 CURRENT = AMOUNT;
 %ADDER;
 END;
 ELSE IF DATE = (&WHEN - 1) THEN DO;
 PRIOR = AMOUNT;
 %ADDER;
 END;
 ELSE DELETE;

IF EOF THEN DO;
 CALL SYMPUT(^
 COUNT',
 LEFT(PUT(COUNT,COMMA5.)));
 CALL SYMPUT(^
 TOTAL',
 LEFT(PUT(TOTAL',DOLLAR5.2)));
 END;

PROC PRINT DATA=REPORT;
 VAR CURRENT PRIOR;
This example generates our monthly claim report much as in Figure 2. Using macro %ADDER, we will accumulate a claim count and total amount only for the records we want. At the end of file, we will pass these sums to TITLE2 in the report. As we are running the program for December of 1997 (i.e. the month is “12”), we will also want to call the backup program (macro %BACKITUP).

The %SUBSTR function allows us to distinguish between the year and month portions of &WHEN. The month value will cause macro %BACKITUP to execute (conditional upon the month being December). The year value is used in conjunction with the %EVAL function to produce begin and end values for selecting the data in the CLAIMS dataset to output. We use the CALL SYMPUT function (technically a DATA step statement) to take DATA step variables (COUNT and TOTAL) and output their values to macro variables of the same names.

Macro %ADDER will bring in a program fragment as needed. This saves a little repetitive typing, although a regular DATA step “LINK” or “GO TO” statement would also have served. The “LEFT” DATA step function in the CALL SYMPUT statement offers a measure of formatting control of the right-justified character value that will otherwise be displayed in TITLE2.

**DELIMITERS**

Along with the special symbols & and %, on occasion, we need to consider delimiters, characters that indicate the end of a macro variable. We have blanks and equal signs (=) above. Ampersands and percent signs (& and %) also operate as delimiters, as does the period (.). In Figure 4 above, note that LIBNAME refers to data set “CLAIMS&YEAR..BACKUP”. The two periods assure us that at macro resolution time, the first period will be the delimiter for macro variable YEAR (and disappear!), while the second will remain as an operating system delimiter in a long data set name.

We will use macro %CENTER to center text strings for a report produced in a DATA _NULL_ step. Several delimiters are illustrated.

**Figure 5 A Few Delimiters**

```
%LET YEAR = 97;
%LET MONTH = 12;

DATA _NULL_;
  FILE PRINT HEADER = H;
  SET CLAIMS;
  /* other SAS statements */
  RETURN;

H:
  PUT %CENTER(FINAL REPORT FOR &MONTH%STRING(&)&YEAR);
  PUT %CENTER(NOTE: BACKUP FILE CREATED &YEAR&MONTH);
  PUT %CENTER(---&YEAR.&MONTH---);
  RETURN;
```

```
%MACRO CENTER(TEXT, WIDTH=132);
  %LET POS = %EVAL((&WIDTH / 2) - (%LENGTH(&TEXT) / 2) / 2));
  @ &POS "&TEXT" /* this is passed */
  /* was that a missing semi-colon */
%MEND CENTER;
```

Figure 5 offers a good example of the use of a program fragment, a section of code that could be written using just
regular SAS statements, but it would be much more messy. It is also a likely candidate to be saved as a program module in standard library at your site. Note the use of the macro function %STR to allow a macro token (the ampersand) to be passed and to serve as a delimiter. We also have two macro parameters in this macro (&TEXT and &WIDTH). We are allowing the latter to default to a value of “132” — the width of a mainframe printer page — but can override this for, lets say, an LRECL of 80 as in this example:

```
PUT %CENTER(REPORT FOR &YEAR, WIDTH=80);
```

**DEBUGGING**

Identifying and fixing macro language errors can be quite a challenge. Good programming practice will help to limit the extent of your task. This includes designing your macro program carefully. Test regular programming steps to confirm that they work (and are not a source of apparent macro errors). Then build and test the macros incrementally, commenting out macro statements irrelevant to your current testing.

As macros get more complicated, be sure to pay special attention to your referencing environment, i.e., the arena in which macros, macro statements, variables, and values will be resolved. In a simple example in Figure 6 below, we attempt to evaluate the value of macro variable &COUNT before it exists. The resulting message in the log will soon become an old friend.

**Figure 6 A Common Error**

```
%Macro CLAIMSREP;
DATA _NULL_;
   SET CLAIMS END=EOF;
   COUNT = 1;
   IF EOF THEN CALL SYMPUT('COUNT',
      LEFT(PUT(COUNT,5.)));
%if &COUNT > 0 %then %do;
     /* additional SAS statements */
%end;
%Mend CLAIMSREP;

%CLAIMSREP;
```

**WARNING: Apparent symbolic reference COUNT not resolved.**

SAS will try to resolve the macro statement referencing the value of &COUNT before the DATA step—which creates &COUNT and assigns a value—executes. A RUN statement immediately following the DATA step will correct this timing problem.

Liberal use of %PUT statements are a helpful way to keep track of the current value of a macro variable. These operate much as a DATA step “PUT” statement and write the indicated information in the SAS log. The following %PUT statement

```
%put count is &count;
```

results in:

```
COUNT IS 24
```

One useful feature of the %PUT is that it is valid anyplace in your program. Many other macro statements are acceptable only inside a macro, and misuse will leave you with a log message such as:

```
ERROR: The %if statement is not valid in open code.
```

**MACRO OPTIONS**

Options most frequently used in debugging programs are MPRINT, MLOGIC, and SYMBOLGEN. All three result in enhanced log messages. MPRINT displays the resolved regular SAS code which will execute. MLOGIC traces macro execution. SYMBOLGEN will cause the value of macro variables to be automatically listed in the log. The first iteration of macro CLAIMREP from Figure 3 above (i.e., &YEAR=89) will result in the logs for each option, respectively, as shown in Figure 7.

**Figure 7 Macro Options**

```
Options MPRINT;
MPRINT(CLAIMREP): PROC PRINT
  DATA=CLAIMS;
MPRINT(CLAIMREP): WHERE YEAR = 89;
MPRINT(CLAIMREP): TITLE1 "ANNUAL REPORT FOR 1989";
MPRINT(CLAIMREP): FOOTNOTE "THRU 06JUL98";

**************************;
Options MLOGIC;
MLOGIC(CLAIMREP): Beginning execution.
MLOGIC(CLAIMREP): %GLOBAL NOTE
MLOGIC(CLAIMREP): %LET (variable name is CURYEAR)
MLOGIC(CLAIMREP): %DO loop beginning:
index variable YEAR; start value is 89;
stop value 98; by value is 1.
```

4
MLOGIC(CLAIMREP): %IF condition "&YEAR" = "&CURYEAR" is FALSE
MLOGIC(CLAIMREP): %DO loop index
variable YEAR is now 90; loop will iterate again.
MLOGIC(CLAIMREP): %IF condition "&YEAR" = "&CURYEAR" is TRUE
MLOGIC(CLAIMREP): %LET (variable name is NOTE)

**************************;
OPTIONS SYMBOLGEN;
SYMBOLGEN: Macro variable YEAR resolves to 89
SYMBOLGEN: Macro variable CURYEAR resolves to 98
SYMBOLGEN: Macro variable NOTE resolves to FOOTNOTE “THRU 06JUL98”

Frequently a combination of these options is used in debugging programs. A result, though, can be very lengthy and busy logs. Once a program is complete, you may want to turn these options off with:

OPTIONS NOMPRINT NOMLOGIC NOSYMBOLGEN;

WHEN MACROS REALLY HELP

The preceding examples have been fairly simple. To help illustrate the power of the macro language, we will first examine a simple utility macro in Figure 8. Based on the number of incoming non-blank drug codes, we will calculate the number of lines to reserve for footnote decodes at the bottom of each page of the report. The maximum number of drug codes, a function of the protocol, would normally be established earlier. In this example we will use a %LET statement to assign a value to macro variable &MAX.

Figure 8 A Simple Utility Macro

/* concatenate drug code values */
%MACRO CONCAT(VAR_NAME,DIMENSN);
%DO I = 1 %TO &DIMENSN;
   &VAR_NAME&I  ||
%END;
%MEND CONCAT;

/* calculate lines to reserve */
%GLOBAL RESERVE;
%LET MAX = 20;
DATA _NULL_; 
   SET DRUGS(KEEP=DRUG1-DRUG&MAX)
   END=EOF;
   ALL = COMPRESS(%CONCAT(DRUG,&MAX)');
   IF ALL NE '' THEN 
       RESERVE = CEIL(LENGTH(ALL) / 3);
   ELSE RESERVE = 0;
   IF EOF THEN 
       CALL(SYMPUT('RESERVE',RESERVE);
RUN;

A somewhat more complex use of macros in Figure 9 is designed to achieve results which would be considerably harder without the use of the macro language. We seek to capture information regarding the distribution of a test statistic for a particular sample where determining the appropriate function and then calculating the actual distribution of the statistic would be quite difficult. By accumulating PROC MEANS output for each of 99 iterations of a sample, we may instead simply measure the nature of our final dataset.

Figure 9 A Not So Simple Set of Macros

/* the analysis/sample dataset */
DATA TRPK;
   SET TRPEAK;
   RENAME CTR = Y0
          CPK = Y1;
%GLOBAL T FIRST CONTIN;
%LET CONTIN = YES;
%MACRO TTEST;

/* create PROC MEANS output */
DATA ONE;
   SET TRPK;
   ZT = Y0 - Y1 * (&T / 100);
   PROC MEANS DATA = ONE T PRT;
   VAR ZT;
   OUTPUT OUT=T&T T=T PRT=PRVALUE;

/* if first iteration of sample */
%IF FIRST = YES %THEN %DO;
   DATA ALLT;
   SET T&T;
   IF .05 <= PVALUE <= .95;
       ITERAT = &T;
   RUN;
%END;

/* all subsequent iterations */
%ELSE %DO;

DATA ALLT;
   SET ALLT T&T(IN=LATEST);
   IF LATEST THEN ITERAT = &T;
%MEND TTEST;

**************************;
%MACRO REPORT;
   /* select title */
   %IF "&CONTIN" = "NO"
   %THEN %LET TITLE = DISCONTINUOUS;
   %ELSE %LET TITLE = CONTINUOUS (MIN IS &MIN, MAX IS &MAX);

   /* report results */
   PROC PRINT DATA=ALLT LABEL;
   VAR ITERAT T PVALUE;
   LABEL T = 'T STATISTIC'
         PVALUE = 'P VALUE'
         ITERAT = 'LOOP ITERATION';
   TITLE "THE DISTRIBUTION IS &TITLE";
%MEND REPORT;

**************************;
%MACRO CONTROL;
   /* call TTEST macro 99 times */
%DO T = 1 %TO 99;
   %IF FIRST = %THEN %LET FIRST = YES;
   %ELSE %LET FIRST = NO;
   %TTEST; /* call TTEST macro */
%END;

/* test continuity, capture min & max */
DATA _NULL_;
   SET ALLT END=EOF;
   RETAIN MIN 100 MAX 0 LAST 0;
   IF _N_ = 1 THEN LAST = ITERAT;
   ELSE DO;
      IF ITERAT NE (LAST + 1) THEN DO;
         CALL SYMPUT('CONTIN','NO');
         STOP;
      END;
      ELSE LAST = ITERAT;
   END;
   IF ITERAT < MIN THEN MIN = ITERAT;
   IF ITERAT > MAX THEN MAX = ITERAT;
   IF EOF THEN DO;
      CALL SYMPUT('MIN',MIN);
      CALL SYMPUT('MAX',MAX);
   END;
%REPORT; /* call REPORT macro */
%MEND CONTROL;
%CONTROL; /* call CONTROL macro */

The %CONTROL macro manages (controls) the work flow. Macro %TTEST is called for 99 iterations of the statistic-generating step. The resulting output is combined in dataset ALLT and after the last iteration, passed back to %CONTROL. The nature of the resulting distribution is measured and placed in macro variables &CONTIN, &MIN, and &MAX. These variables are processed further when macro %REPORT is called to determine the appropriate TITLE statement. This is then printed along with the data.

SAS is a registered trademark of the SAS Institute, Inc., Cary, NC, USA.

SUGGESTED READING

The macro language has proven exceedingly popular and any of the SAS user conferences’ proceedings will most likely contain several papers on this topic. The SAS manual is another good resource. A very short list of references might include the following:


Jeff Phillips, Veronica Walgamotte, and Derek Drummond, Warning: Apparent Macro Invocation not Resolved. . . Techniques for Debugging Macro


Author Contact:
John J. Cohen
Jcohen1265@aol.com